Math 420/620 Quiz 10

1. A squadron of 16 bombers needs to penetrate air defenses to reach its target. They can either fly low and expose themselves to the air defense guns, or fly high and expose themselves to surface–to–air missiles. In either case, the air defense firing sequence proceeds in three stages. First, they must detect the target, then they must acquire the target (lock on target), and finally they must hit the target. Each of these stages may or may not succeed. The probabilities are as follows:

AD Type	Pdetect	Pacquire	Phit
Low	0.90	0.80	0.05
High	0.75	0.95	0.70

The guns can fire 20 shells per minute, and the missile installation can fire three per minute. The proposed flight path will expose the planes for one minute if they fly low, and five minutes if they fly high.

(i) Determine the optimal flight path (low or high). The objective is to maximize the bombers that survive to strike the target.

For the low flying planes define

- $t_L = 1$ the time in minutes the low-flying plane is exposed to defensive fire.
- $r_L = 20$ the number of shells fired on low-flying planes per minute.
- p_L the probability a shell hits a low-flying plane.
- S_L the number of low-flying planes that survive.

and for the high-flying planes

- $t_H = 5$ the time in minutes the high-flying plane is exposed to defensive fire.
- $r_H = 3$ the number of missles fired on high-flying planes per minute.
- p_H the probability a missle hits a high-flying plane.
- S_H the number of high-flying planes that survive.

We assume that the probabilities are independent so that

$$q_a = P_{\text{detect}} \cdot P_{\text{acquire}} \cdot P_{\text{hit}}$$

where the altitude $a \in \{L, H\}$ and the values for $P_{\text{detect}}, P_{\text{acquire}}$ and P_{hit} are given by the corresponding line in the table.

In [1]:
```
pL=0.90*0.80*0.05
```

Out[1]: 0.036000000000000004

Since the enemy doesn't fire at planes which are already shot down, every shell is assumed to be shot at a plane that is still flying. If the defenders run out of planes to fire at they stop shooting. Over t_L minutes a total of $t_L r_L = 20$ shells are fired.

In [2]:
```
tL=1; rL=20; tL*rL
```

Out[2]: 20

Let X_i be a random variable equal to 1 if a shell number i hits the target and 0 if it misses. Assuming each shell is independently targeted, we have

$$X_i = \begin{cases} 1 & \text{with probability } p_L \\ 0 & \text{with probability } 1 - p_L \end{cases} \qquad \text{for} \qquad i = 1, 2, \ldots, 20.$$

If there were an infinite number of planes, the total number of low-flying planes shot down would be

$$T_L = \sum_{i=1}^{20} X_i.$$

Let S_L be how many out of the 16 planes survive. Therefore

$$S_L = \begin{cases} 0 & \text{with probability } \mathbf{P}\{T_L \geq 16\} \\ n & \text{with probability } \mathbf{P}\{T_L = 16 - n\}. \end{cases}$$

Here $n = 1, 2, \ldots, 16$. The expected number that survive is

$$\mathbf{E}[S_L] = \sum_{n=1}^{16} n \mathbf{P}\{T_L = 16 - n\}.$$

Since T_L is a sum of independent Bernoulli random variables it has a binomial distribution. Therefore

$$\mathbf{P}\{T_L = m\} = \binom{20}{m} p_L^m (1 - p_L)^{20-m}.$$

It follows that

$$\mathbf{E}[S_L] = \sum_{n=1}^{16} n \binom{20}{16-n} p_L^{16-n} (1 - p_L)^{4+n}$$

In [3]:
```
PTL(m)=binomial(20,m)*pL^m*(1-pL)^(20-m)
```

```
ESL=sum(n*PTL(16-n) for n=1:16)
```

Out[3]: 15.279999999999999

Note that an altermate way to approximate the same value is to assume the probability that all planes are shot down is vanishingly small.

In [4]:
```
QTL16=sum(PTL(n) for n=17:20)
```

Out[4]: 2.9443114378667005e-22

Thus, $P\{T_L > 16\} \approx 2.9443114378667005 \times 10^{-22}$ implies

$$\mathbf{P}\{T_L \leq 16\} = 1 - P\{T_L > 16\} \approx 1$$

Since

$$S_L = \begin{cases} 0 & \text{for } T_L > 16 \\ 16 - T_L & \text{for } T_L \leq 16, \end{cases}$$

it follows that

$$\mathbf{E}[S_L] = \mathbf{E}[16 - T_L]\mathbf{P}\{T_L \leq 16\} \approx 16 - \mathbf{E}[T_L].$$

In [5]:
```
20*pL
```

Out[5]: 0.7200000000000001

In [6]:
```
16-20*pL
```

Out[6]: 15.28

Computing

$$\mathbf{E}[T_L] = \sum_{i=1}^{20} \mathbf{E}[X_i] = 20p_L \approx 0.72$$

yields that

$$\mathbf{E}[S_L] \approx 15.28.$$

Note that in spite of dropping the small term, numerically, this turns out to the same as the previously computed answer.

Now perform a similar analysis for the high-flying planes.

In [7]:
```
pH=0.75*0.95*0.70
```

Out[7]: 0.49874999999999999

In [8]:
```
tH=5; rH=3; tH*rH
```

Out[8]:
```
15
```

Let Y_i be a random variable equal to 1 if a missle number i hits the target and 0 if it misses. Assuming each missle is independently targeted, we have

$$Y_i = \begin{cases} 1 & \text{with probability } p_H \\ 0 & \text{with probability } 1-p_H \end{cases} \qquad \text{for} \qquad i = 1, 2, \ldots, 15.$$

If there were an infinite number of planes, the total number of high-flying planes shot down would be

$$T_H = \sum_{i=1}^{15} Y_i.$$

Let S_H be how many out of the 16 planes survive. Therefore

$$S_H = \begin{cases} 0 & \text{with probability } 0 \\ n & \text{with probability } \mathbf{P}\{T_H = 16-n\}. \end{cases}$$

Here $n = 1, 2, \ldots, 16$. The expected number that survive is

$$\mathbf{E}[S_H] = \sum_{n=1}^{16} n\mathbf{P}\{T_H = 16-n\}.$$

Since T_H is a sum of independent Bernoulli random variables it has a binomial distribution. Therefore

$$\mathbf{P}\{T_H = m\} = \binom{15}{m} p_H^m (1-p_H)^{15-m}.$$

It follows that

$$\mathbf{E}[S_H] = \sum_{n=1}^{16} n \binom{20}{16-n} p_L^{16-n} (1-p_L)^{n-1}$$

In [9]:
```
PTH(m)=binomial(15,m)*pH^m*(1-pH)^(15-m)
ESH=sum(n*PTH(16-n) for n=1:16)
```

Out[9]:
```
8.518750000000002
```

The alternative method of computing $\mathbf{E}[S_H]$ is even easier and exact in this case because there is no chance that all the planes get shot down by missiles.

In [10]:
```
15*pH
```

Out[10]: 7.481249999999998

In [11]:
```
16-15*pH
```

Out[11]: 8.51875

It follows that

$$\mathbf{E}[S_H] = 16 - \mathbf{E}[T_H] = 16 - 15 * p_H \approx 8.51875$$

which is the same as computed above.

In summary

$$\mathbf{E}[S_L] = 15.28 \qquad \text{and} \qquad \mathbf{E}[S_H] = 8.519.$$

Therefore, the optimal path is to fly low so that on average 15.28 survive to strike the target.

(ii) Each individual bomber has a 70 percent chance to destroy the target. Use the results of part (i) to determine the chances of success (target destroyed) for this mission.

Assume success means destroying the target. Let

- $p_D = 0.7$ be the probability that an individual plane destroys the target.

We want to compute

$$\mathbf{P}\{\text{success}\} = \sum_{n=1}^{16} \mathbf{P}\big\{\text{success}|S_L = n\big\}\mathbf{P}\{S_L = n\}$$

Suppose $S_L = n$ planes get through the defenses. Let Z_i be a random variable equal to 1 if plane i hits the target and 0 if not. Assuming each of the planes which get through have an independent chance of success, we have

$$Z_i = \begin{cases} 1 & \text{with probability } p_D \\ 0 & \text{with probability } 1 - p_D. \end{cases} \qquad \text{for} \qquad i = 1, 2, \ldots, n.$$

We want to compute the probability that at least one plane which gets through also hits the target. Thus

$$\mathbf{P}\big\{\, Z_i = 1 \text{ for at least one } i|S_L = n \,\big\} = 1 - \mathbf{P}\big\{\, Z_i = 0 \text{ for all } i|S_L = n \,\big\}.$$

By independence

$$\mathbf{P}\big\{\, Z_i = 0 \text{ for all } i|S_L = n \,\big\} = \prod_{i=1}^{n} \mathbf{P}\{Z_i = 0\} = (1 - p_D)^n.$$

Therefore

$$\mathbf{P}\Big\{\text{success}\big|S_L = n\Big\} = 1 - (1 - p_D)^n$$

combined with the fact from earlier that

$$\mathbf{P}\{S_L = n\} = \binom{20}{16-n} p_L^{16-n}(1-p_L)^{4+n}$$

yields

$$\mathbf{P}\{\text{success}\} = \sum_{n=1}^{16} \big(1 - (1-p_D)^n\big) \binom{20}{16-n} p_L^{16-n}(1-p_L)^{4+n}.$$

In [12]:
```
pD=0.7
```

Out[12]:
```
0.7
```

In [13]:
```
Psuccess=sum((1-(1-pD)^n)*PTL(16-n) for n=1:16)
```

Out[13]:
```
0.9999999783964197
```

Therefore, there is a 99.9999978 percent chance of mission success.

(iii) Determine the minimum number of bombers necessary to guarantee a 95% chance of mission success.

In [14]:
```
Ps(m)=sum((1-(1-pD)^n)*PTL(m-n) for n=1:m)
```

Out[14]:
```
Ps (generic function with 1 method)
```

In [15]:
```
using PrettyTables
T=[1:16 Ps.(1:16)]
pretty_table(T,
    formatters=[fmt__printf("%3d",[1]),fmt__printf("%.6f",[2])],
    column_labels=["planes","P(success)"])
```

planes	P(success)
1	0.336232
2	0.688229
3	0.882921
4	0.961291
5	0.987971
6	0.996353
7	0.998903
8	0.999671
9	0.999901
10	0.999970
11	0.999991
12	0.999997
13	0.999999
14	1.000000
15	1.000000
16	1.000000

It follows that 4 airplanes leads to a 96% chance of mission success but 3 only provides a 88% chance of success. Thus, 4 is the minimum necessary to guarantee at least a 95% chance of success.

(iv) Perform a sensitivity analysis with respect to the probability $p = 0.7$ that an individual bomber can destroy the target. Consider the number of bombers that must be sent to guarantee a 95% chance of mission success.

Let $N(p)$ be the minimum number of planes needed to guarantee a 95% chance of mission success. Since this is a discrete function, we plug in different values for p, recompute the table in the previous part and then automatically search for the minimum number of planes in a squadron for which $\mathbf{P}\{\text{success}\} \geq 0.95.$

In [16]:
```
Ps(p,m)=sum((1-(1-p)^n)*PTL(m-n) for n=1:m)
Tp(p)=[Ps(p,m) for m=1:16]
N95(p)=findfirst(x->x>0.95,Tp(p))
```

Out[16]:
```
N95 (generic function with 1 method)
```

In [17]:
```
# Check that 0.7 still returns 4 planes
N95(0.7)
```

Out[17]:
```
4
```

In [18]:
```
using Plots
```

In [19]:
```
plot(N95,0.18:0.0005:1.0,title="Planes needed for 95 Chance of Success",
    xlabel="Probabilty p to hit target",ylabel="N planes",legend=false)
```

Out[19]:



In [20]:
```
T2=[0.2:0.1:1.0 N95.(0.2:0.1:1.0)];
```

In [21]:
```
pretty_table(T2,
    formatters=[fmt__printf("%.1f",[1]),fmt__printf("%d",[2])],
    column_labels=["p","N(p)"])
```

p	N(p)
0.2	15
0.3	10
0.4	7
0.5	6
0.6	5
0.7	4
0.8	4
0.9	4
1.0	3

(iv) Bad weather reduces both P_{detect} and p, the probability that a bomber can destroy the target. If all of these probabilities are reduced in the same proportion, which side gains an advantage in bad weather?

Let γ be the proportion by which the probabilities are reduced. Write γp_L and γp_D for the new probabilities that a shell hits the plane and the new probability that a surviving plane hits the target. As a function of γ the probability of success is now

$$\mathbf{P}\{\text{success}\} = \sum_{n=1}^{16} \left(1-(1-\gamma p_D)^n\right)\binom{20}{16-n}(\gamma p_L)^{16-n}(1-\gamma p_L)^{4+n}.$$

In [22]:
```
# The probability P(TL=m) when gamma=g
PTLg(g,m)=binomial(20,m)*(g*pL)^m*(1-g*pL)^(20-m)
# Expected number of planes remaining with gamma=g and m planes
ESLg(g,m)=sum(n*PTLg(g,m-n) for n=1:m)
# The probability of success with gamma=g, pD=p and m planes
Psg(g,p,m)=sum((1-(1-g*p)^n)*PTLg(g,m-n) for n=1:m)
```

Out[22]:
```
Psg (generic function with 1 method)
```

In [23]:
```
plot(g->Psg(g,0.7,16),0.05:0.01:1.0,title="Vary p with N fixed",
    ylabel="P(success)",xlabel="gamma",label="p=0.7 N=16")
plot!(g->Psg(g,0.4,16),0.05:0.01:1.0,
    ylabel="P(success)",xlabel="gamma",label="p=0.4 N=16")
plot!(g->Psg(g,0.2,16),0.05:0.01:1.0,
    ylabel="P(success)",xlabel="gamma",label="p=0.2 N=16")
```

Out[23]:



In [24]:
```
plot(g->Psg(g,0.7,16),0.05:0.01:1.0,title="Fix p and vary N",
    ylabel="P(success)",xlabel="gamma",label="p=0.7 N=16")
plot!(g->Psg(g,0.7,8),0.05:0.01:1.0,
    ylabel="P(success)",xlabel="gamma",label="p=0.7 N=8")
plot!(g->Psg(g,0.7,4),0.05:0.01:1.0,
    ylabel="P(success)",xlabel="gamma",label="p=0.7 N=4")
```

Out[24]:



In all cases the curves are increasing functions of γ. This means the probability of mission success $\mathbf{P}\{\text{success}\}$ decreases as the weather gets worse. Therefore, bad weather is advantageous to the defender.

In [ ]: